Newsgroup: it.comp.lang.visual-basic
Titolo: IMMAGINI TRASPARENTI

Come faccio a fare in modo che vb consideri lo sfondo di un'immagine
"trasparente"??

Potresti creare un'immagine gif con qualche editor di gif animate.
Ciao
d.b.a. Andrea Pauselli - Pietralunga
--
Posted from 
velthur.krenet.it
 [194.143.128.45]
via Mailgate.ORG Server - 
http://www.mailgate.org

 "trasparente"??
Puoi crearti un' ocx apposito. Basta giocare con le sue proprietą
MaskPicture,Picture,MaskColor per rendere un' immagine trasparente.
Se hai bisogno di lucidazione ti posso postare il codice occorrente.
Ciao
Daniele
---------------------------------------------------------------------------
Le risposte alle tue domande le trovi sul Sito Comune di questo NewsGroup:
 
http://www.murialdo.it/it_lang_vb
---------------------------------------------------------------------------

...


 lo potresti mandare a it.binari.visual-basic oppure postarmelo????
Visto che altre richieste mi sono pervenute via email e che il codice non è
molto lungo, credo sia il caso di postarlo sul NG (chiedo cmq scusa a chi
non è interessato per la lunghezza del post)
L'utilizzo è abbastanza intuitivo, cmq la picture da visualizzare la devi
passare alla proprietà Immagine (se: set
NomeIstanza.immagine=loadpicture("pippo.bmp").
La proprietà MaskColor è il colore che sarà reso trasparente.
Per qualsiasi altro chiarimento contattemi.
Ciao
Daniele
Apri un progetto Controllo ActiveX e ci incolli questo codice...
Option Explicit
Dim Auto As Boolean
Public Event Click()
Public Event DblClick()
Public Property Get Immagine() As Picture
Set Immagine = UserControl.Picture
End Property
Public Property Set Immagine(ByVal vNewValue As Picture)
If Auto = True Then
UserControl.Width = UserControl.ScaleX(vNewValue.Width, vbHimetric,
UserControl.ScaleMode)
UserControl.Height = UserControl.ScaleY(vNewValue.Height, vbHimetric,
UserControl.ScaleMode)
End If
UserControl.Picture = vNewValue
UserControl.MaskPicture = vNewValue
End Property
Public Property Get MaskColor() As OLE_COLOR
MaskColor = UserControl.MaskColor
End Property
Public Property Let MaskColor(ByVal vNewValue As OLE_COLOR)
UserControl.MaskColor = vNewValue
End Property
Private Sub UserControl_Click()
RaiseEvent Click
End Sub
Private Sub UserControl_DblClick()
RaiseEvent DblClick
End Sub
Private Sub UserControl_Initialize()
UserControl.BackStyle = 0
End Sub
Private Sub UserControl_ReadProperties(PropBag As PropertyBag)
On Error Resume Next
Set Immagine = PropBag.ReadProperty("Immagine")
MaskColor = PropBag.ReadProperty("MaskColor", RGB(0, 0, 0))
Enabled = PropBag.ReadProperty("Enabled", True)
AutoSize = PropBag.ReadProperty("AutoSize", False)
End Sub
Public Property Get Enabled() As Boolean
'Ricordati di assegnare l'Id di routine
Enabled = UserControl.Enabled
End Property
Public Property Let Enabled(ByVal vNewValue As Boolean)
'Ricordati di assegnare l'Id di routine
UserControl.Enabled = vNewValue
End Property
Private Sub UserControl_WriteProperties(PropBag As PropertyBag)
PropBag.WriteProperty "Immagine", UserControl.Picture
PropBag.WriteProperty "MaskColor", UserControl.MaskColor, RGB(0, 0, 0)
PropBag.WriteProperty "Enabled", UserControl.Enabled, True
PropBag.WriteProperty "AutoSize", Auto, False
End Sub
Public Sub Refresh()
UserControl.Refresh
End Sub
Public Property Get AutoSize() As Boolean
AutoSize = Auto
End Property
Public Property Let AutoSize(ByVal vNewValue As Boolean)
Auto = vNewValue
If Auto = True Then
UserControl.Width = UserControl.ScaleX(UserControl.Picture.Width,
vbHimetric, UserControl.ScaleMode)
UserControl.Height = UserControl.ScaleY(UserControl.Picture.Height,
vbHimetric, UserControl.ScaleMode)
End If
End Property
